
#!/usr/bin/gnuplot
#
# nicht invasive Bludtuckmessung, Ablassrate Detail
#
#      gnuplot  NIBD_Ablassrate.gnuplot -
#   
# Waagrechte, X Achse bildet die Zeit ab.  
set xdata time
#
## Die Zeitangaben werden im ISO Format, 
# "http://de.wikipedia.org/wiki/ISO_8601" einglesen,
# die kleinste Auflösung ist eine Sekunde
#set timefmt x "%Y-%m-%d %H:%M:%S"
set timefmt '"%M:%S"'
#
# Bereichsgrenzen der x Achse
set xrange ['"00:00"':'"00:33"']
#
# Achsenbeschriftung x
set xlabel "Zeit"
#
# Bereichsgrenzen der y Achse
set yrange [0 : 200] 
#
# Achsenbeschriftung y
set ylabel "mmHg" ."\n" ."1/min"
#
set grid ytics mytics  # draw lines for each ytics and mytics
set grid xtics mytics  # draw lines for each xtics and mytics
set mytics 5           # set the spacing for the mytics
set grid               # enable the grid
#
# Datenpunkte
set sample 901
#
# Kreisfrequenz, Winkelfrequenz als Funktion von x
omega(x)=(2*pi*x)
#
# Variablen
Puls=64
_60_sek_je_min=60.0  #Dezimalpunkt für Gleitkommazahl
Start_Winkel=pi*2/3
Amplitude=20
Versatz=100
#
# Startdruck
Initialdruck=170
#
# Ablassrate als lineare Funktion von x
Ablassrate_3mmHg_je_Sekunde(x)= -3*x+Initialdruck
Ablassrate_6mmHg_je_Sekunde(x)= -6*x+Initialdruck
#
# Bildschimausgabe
plot Versatz+Amplitude*cos((Puls/_60_sek_je_min)*omega(x)+Start_Winkel), \
     Ablassrate_3mmHg_je_Sekunde(x), \
     Ablassrate_6mmHg_je_Sekunde(x)
#
# Dateiausgabe für Druck,…
#
# Dateiname für die Ausgabe als scalable vector graphik (svg).
fname="NIBD_Ablassrate"
#
########################################################################
### svgfile = fname + .svg, Datei erzeugen gleich wie Bildschirm zu sehen ist
svgfile=fname .".svg"
set terminal push      # Speichere die aktuellen terminal Einstellungen
### Setze das terminal als SVG und Größe 1/3 verkleinert für A4 passed.
set terminal svg size 2970/3, 2100/3  d
set output svgfile # SVG Datei angeben.
replot             # SVG Datei ausgeben
set terminal wxt   # Terminal zurucksetzen
set output         # zurückholen des interaktiven Terminals
set terminal pop   # zuruckholen der Einstellunge des interaktiven Terminals
########################################################################


